
//***************************************************************
//* MACRO:  WATERMRK.WCM
//* PURPOSE:  Helps user create a watermark from a .wpg file or from text.
//***************************************************************
Application (A1; "WordPerfect"; Default; "UK")

If(?Substructure)
MessageBox(; "Not At Main Edit Screen"; "This macro can be played only from the main
editing screen.  Exit all substructures (headers, text boxes, etc.) and play the macro 
again."; IconStop!)
Quit

EndIf

DefaultFontSize:=48p // Default font size
WtmkA:=1 // Default to use watermark A
WpgWtmk:=1 // Default to graphic watermark
Path:=?PathGraphics
If(Path="")

Path:=?PathProgram
Call(CheckDir@)

EndIf
If(SubStr(Path; StrLen(Path); 1)<>"\")

Path:=Path+"\"
EndIf
HeightOfBox:=?PaperLength
WidthOfBox:=?PaperWidth

Label(MainDialog@)
DialogDefine("Dlg1"; 50; 50; 157; 85; 1+2+16; "Watermark Options")
DialogAddGroupBox("Dlg1"; "Gb2"; 8; 4; 56; 38; "T&ype")
DialogAddRadioButton("Dlg1"; "Rb4"; 18; 14; 36; 11; "&Text"; TxtWtmk)
DialogAddRadioButton("Dlg1"; "Rb5"; 18; 26; 36; 11; "&Graphic"; WpgWtmk)
DialogAddGroupBox("Dlg1"; "Gb3"; 72; 4; 72; 38; "&Placement")
DialogAddRadioButton("Dlg1"; "Rb6"; 82; 14; 53; 11; "Watermark &A"; WtmkA)
DialogAddRadioButton("Dlg1"; "Rb7"; 82; 26; 53; 11; "Watermark &B"; WtmkB)
Display(On!)
DialogDisplay("Dlg1"; 1)
DialogResult:=MacroDialogResult
DialogDestroy("Dlg1")
Display(Off!)
If(DialogResult=2)

Go(End@)
EndIf

If(TxtWtmk=1)
Call(TextWatermark@)



Else
Call(GraphicWatermark@)

EndIf

// Do this only at the main document screen
While(?Substructure<>0)
SubstructureExit()
EndWhile
MacroStatusPrompt(On!; "Placing watermark...")
If(WtmkA=1)

If(?WatermarkAOccur<>0)
StructureName:="Watermark A"
Call(StructureExists@)

EndIf
WatermarkA(Create!)

Else
If(WtmkB=1)

If(?WatermarkBOccur<>0)
StructureName:="Watermark B"
Call(StructureExists@)

EndIf
WatermarkB(Create!)

EndIf
EndIf

If(TxtWtmk=1)
CenterCurrentPage(On!)
Justification(Center!)
FontSize(DefaultFontSize)
Type(WatermarkText)
Display(On!)
MessageBox(MsgBoxResult; "Watermark Text Options"; "The current font is ^0, and the 
current size is ^1.  Would you like to change these options?"; IconQuestion! | YesNo! | 
HasParameters!; ?Font; DefaultFontSize)
Display(Off!)
If(MsgBoxResult=6)

PosDocTop()
FontDlg()

EndIf
Else

BoxCreate(WatermarkImageBox!)
BoxAttachTo(Page!)
BoxPageOffset(No!)
BoxHorizontalPosition(0.0")
BoxVerticalPosition(0.0")
BoxHeight(HeightOfBox)



BoxWidth(WidthOFBox)
BoxBorder(0)
BoxContentType(Image!)
BoxImageRetrieve(MakeInternal!; Path+Wtrmk[Count; 1])
BoxEnd(Save!)

EndIf
SubstructureExit ()

Label(End@)
Quit

//***************************************************************
//* ROUTINE:  TextWatermark@
//* INPUT VARIABLES:  none
//* OUTPUT VARIABLES:  
//* DESCRIPTION:  Displays a dialog for entering watermark text.
//***************************************************************
Label(TextWatermark@)
DialogDefine("Dlg1"; 50; 50; 172; 131; 1+2+16; "Watermark Creation Macro")
DialogAddEditBox("Dlg1"; "Eb1"; 8; 8; 150; 50; 64+32; WatermarkText; 1024)
DialogAddHLine("Dlg1"; "Hl1"; 6; 66; 154)
DialogAddText("Dlg1"; "T1"; 8; 70; 150; 20; 1; "Enter text to place in the watermark.  Use Ctrl-
Enter to start a new line.")
Display(On!)
DialogDisplay("Dlg1"; "Eb1")
DialogResult:=MacroDialogResult
DialogDestroy("Dlg1")
Display(Off!)
If(DialogResult=2)

Go(End@)
EndIf
Return
//***************************************************************

//***************************************************************
//* ROUTINE:  GraphicWatermark@
//* INPUT VARIABLES:  none
//* OUTPUT VARIABLES:  
//* DESCRIPTION:  Displays a dialog for selecting a watermark graphic.
//***************************************************************
Label(GraphicWatermark@)
Call(Definitions@)
MacroStatusPrompt(On!; "Initialising dialog...")
DialogDefine("Dlg1"; 50; 50; 132; 130; 1+2+16; "Graphic Watermarks")
DialogAddListBox("Dlg1"; "Lb1"; 8; 8; 110; 78; 1; Choice)
For(Count; 1; Count<=MaxWtrmk; Count+1)



DialogAddListItem("Dlg1"; "Lb1"; Wtrmk[Count; 2])
EndFor
DialogAddPushButton("Dlg1"; "Pb1"; 6; 94; 38; 14; 0; "P&review")
DialogHandle(hwndLb1; "Dlg1"; "Lb1")
DLLLoad(UserLink; "User")

// Select the first item
DLLCall(UserLink; "SendMessage"; nVoid:INTEGER;{LoWord(hwndLb1); LoWord(1031); 
LoWord(0); 0})
DllFree(UserLink)
MacroStatusPrompt(Off!)
Label(GraphicDialogShow@)
Display(On!)
DialogDisplay("Dlg1"; "Lb1")
DialogResult:=MacroDialogResult
Display(Off!)

If (DialogResult =2)
DialogDestroy("Dlg1")
Go(End@)

EndIf

For(Count; 1; Wtrmk[Count; 2]<>Choice; Count+1)
EndFor

FileExists(FileExist; Path+Wtrmk[Count; 1])
If(Not FileExist)

MessageBox(MsgBoxResult; "Graphic Not Found"; "^0 does not exist.  Check your 
graphics directory in preferences, or press Retry to select a different directory."; 
IconExclamation! | RetryCancel! | HasParameters! | Beep!; Path+Wtrmk[Count; 1])
If(MsgBoxResult=4)

Call(CheckDir@)
EndIf
Go(GraphicDialogShow@)

EndIf
If(DialogResult="Pb1")

Call(ViewDlg@)
Go(GraphicDialogShow@)

EndIf
DialogDestroy("Dlg1")
Return
//***************************************************************

//***************************************************************
//* ROUTINE:  StructureExists@
//* INPUT VARIABLES:  StructureName
//* OUTPUT VARIABLES:  



//* DESCRIPTION:  Displays a message box.
//***************************************************************
Label(StructureExists@)
MessageBox(MsgBoxResult; StructureName+" Already Exists"; "^0 already exists.  If you 
continue, the new watermark will replace the current one."; IconExclamation! | OKCancel! | 
HasParameters!; StructureName)
If(MsgBoxResult<>1)

Go(MainDialog@)
EndIf
Return
//***************************************************************

//***************************************************************
//* ROUTINE:  ViewDlg@
//* INPUT VARIABLES:  Path; Wtrmk[Count; 1]
//* OUTPUT VARIABLES:  None
//* DESCRIPTION:  Displays a viewer for the highlighted file.
//***************************************************************
Label(ViewDlg@)
DlgHeight:=176
DlgWidth:=116
DialogDefine("Dlg2"; 50; 50; DlgWidth+20; DlgHeight+50; 1+16; "Preview Watermark")
DialogAddViewer("Dlg2"; "V1"; 7; 8; DlgWidth; DlgHeight; Path+Wtrmk[Count; 1])
DialogDisplay("Dlg2"; 1)
DialogDestroy("Dlg2")
Return
//***************************************************************

//***************************************************************
//* ROUTINE:  CheckDir@
//* INPUT VARIABLES:  Path
//* OUTPUT VARIABLES:  Path
//* DESCRIPTION:  This will create a dialog to prompt for a directory.
//***************************************************************
Label(CheckDir@)
DialogDefine("PathDlg"; 50; 50; 172; 90; 1+2+16; "Graphic Directory")
DialogAddFilenameBox("PathDlg"; "Fb1"; 8; 8; 150; 14; 1; Path; Path; "*.wpg")
DialogAddText("PathDlg"; "T1"; 8; 28; 150; 20; 1; "Please select the directory that contains the 
Watermark graphics files.")
Label(Redisplay@)
Display(On!)
DialogDisplay("PathDlg"; "Fb1")
Display(Off!)
If(MacroDialogResult=2)

DialogDestroy("PathDlg")
Go(End@)



EndIf
If(Path="")

Go(Redisplay@)
EndIf
DialogDestroy("PathDlg")
If(SubStr(Path; StrLen(Path); 1)<>"\")

Path:=Path+"\"
EndIf
Return
//***************************************************************

//***************************************************************
//* ROUTINE:  Definitions@
//* INPUT VARIABLES:  None
//* OUTPUT VARIABLES:  WtrMk[]
//* DESCRIPTION:  Defines the array for the page borders.
//***************************************************************
Label(Definitions@)
MaxWtrmk:=30 // This must be increased if more are added
Declare(Wtrmk[MaxWtrmk; 3])

// Format for border definitions:     
Wtrmk[01; 1]:="overdue.WPG" // Filename
Wtrmk[01; 2]:="Account Overdue" // Descriptive name
Wtrmk[02; 1]:="approved.WPG"
Wtrmk[02; 2]:="Approved"
Wtrmk[03; 1]:="asap.WPG"
Wtrmk[03; 2]:="A.S.A.P"
Wtrmk[04; 1]:="certify.WPG"
Wtrmk[04; 2]:="Certified"
Wtrmk[05; 1]:="classifi.WPG"
Wtrmk[05; 2]:="Classified"
Wtrmk[06; 1]:="confiden.WPG"
Wtrmk[06; 2]:="Confidential"
Wtrmk[07; 1]:="confirm.WPG"
Wtrmk[07; 2]:="Confirmation"
Wtrmk[08; 1]:="prohibit.WPG"
Wtrmk[08; 2]:="Copying Prohibited"
Wtrmk[09; 1]:="copyrite.WPG"
Wtrmk[09; 2]:="Copyrighted Material"
Wtrmk[10; 1]:="dontcopy.WPG"
Wtrmk[10; 2]:="Do Not Copy"
Wtrmk[11; 1]:="draft.WPG"
Wtrmk[11; 2]:="Draft"
Wtrmk[12; 1]:="duplicat.WPG"
Wtrmk[12; 2]:="Duplicate"
Wtrmk[13; 1]:="estimate.WPG"



Wtrmk[13; 2]:="Estimate"
Wtrmk[14; 1]:="file_co.WPG"
Wtrmk[14; 2]:="File Copy"
Wtrmk[15; 1]:="final.WPG"
Wtrmk[15; 2]:="Final"
Wtrmk[16; 1]:="internal.WPG"
Wtrmk[16; 2]:="For Internal Use Only"
Wtrmk[17; 1]:="eyesonly.WPG"
Wtrmk[17; 2]:="For Your Eyes Only"
Wtrmk[18; 1]:="importnt.WPG"
Wtrmk[18; 2]:="Important"
Wtrmk[19; 1]:="invitatn.WPG"
Wtrmk[19; 2]:="Invitation"
Wtrmk[20; 1]:="nonegot.WPG"
Wtrmk[20; 2]:="Not Negotiable"
Wtrmk[21; 1]:="past_due.WPG"
Wtrmk[21; 2]:="Past Due"
Wtrmk[22; 1]:="personal.WPG"
Wtrmk[22; 2]:="Personal Copy"
Wtrmk[23; 1]:="proof.WPG"
Wtrmk[23; 2]:="Proof"
Wtrmk[24; 1]:="proposal.WPG"
Wtrmk[24; 2]:="Proposal"
Wtrmk[25; 1]:="reply.WPG"
Wtrmk[25; 2]:="Reply Requested"
Wtrmk[26; 1]:="rsvp.WPG"
Wtrmk[26; 2]:="R.S.V.P."
Wtrmk[27; 1]:="rush.WPG"
Wtrmk[27; 2]:="Rush"
Wtrmk[28; 1]:="secret.WPG"
Wtrmk[28; 2]:="Secret"
Wtrmk[29; 1]:="thanks.WPG"
Wtrmk[29; 2]:="Thank You"
Wtrmk[30; 1]:="topsecrt.WPG"
Wtrmk[30; 2]:="Top Secret"
Return
//***************************************************************


